

# Denotational Design

from meanings to programs

Conal Elliott

LambdaJam 2015

*The purpose of abstraction is not to be vague,  
but to create a new semantic level  
in which one can be absolutely precise.*

- Edsger Dijkstra

# Goals

---

- *Abstractions*: precise, elegant, reusable.
- *Implementations*: correct, efficient, maintainable.
- *Documentation*: clear, simple, accurate.

## Not even wrong

---

Conventional programming is precise only about how, not what.

*It is not only not right, it is not even wrong.*

- Wolfgang Pauli

*Everything is vague to a degree you do not realize  
till you have tried to make it precise.*

- Bertrand Russell

*What we wish, that we readily believe.*

- Demosthenes

# Denotative programming

---

Peter Landin recommended “denotative” to replace ill-defined “functional” and “declarative”.

Properties:

- Nested expression structure.
- Each expression *denotes* something,
- depending only on denotations of subexpressions.

“...gives us a test for whether the notation is genuinely functional or merely masquerading.” (*The Next 700 Programming Languages*, 1966)

# Denotational design

---

Design methodology for “genuinely functional” programming:

- Precise, simple, and compelling specification.
- Informs *use* and *implementation* without entangling them.
- Standard algebraic abstractions.
- Free of abstraction leaks.
- Laws for free.
- Principled construction of correct implementation.

# Overview

---

- Broad outline:
  - Example, informally
  - *Pretty pictures*
  - Principles
  - More examples
  - Reflection
- Discussion throughout
- Try it on.

# Example: image synthesis/manipulation

---

- How to start?
- What is success?



# Functionality

- Import & export
- Spatial transformation:
  - Affine: translate, scale, rotate
  - Non-affine: swirls, lenses, inversions, ...
- Cropping
- Monochrome
- Overlay
- Blend
- Blur & sharpen
- Geometry, gradients, ....

# API first pass

**type** *Image*

*over* :: *Image* → *Image* → *Image*

*transform* :: *Transform* → *Image* → *Image*

*crop* :: *Region* → *Image* → *Image*

*monochrome* :: *Color* → *Image*

-- shapes, gradients, etc.

*fromBitmap* :: *Bitmap* → *Image*

*toBitmap* :: *Image* → *Bitmap*

# How to implement?

---

*wrong first question*

## What to implement?

---

- What do these operations mean?
- More centrally: What do the *types* mean?

# What is an image?

---

Specification goals:

- Adequate
- Simple
- Precise

Why these properties?

# What is an image?

---

My answer: assignment of colors to 2D locations.

How to make precise?

**type** *Image*

Model:

$$\mu :: \textit{Image} \rightarrow (\textit{Loc} \rightarrow \textit{Color})$$

What about regions?

$$\mu :: \textit{Region} \rightarrow (\textit{Loc} \rightarrow \textit{Bool})$$

## Specifying *Image* operations

---

$\mu$  (*over top bot*)  $\equiv \dots$

$\mu$  (*crop reg im*)  $\equiv \dots$

$\mu$  (*monochrome c*)  $\equiv \dots$

$\mu$  (*transform tr im*)  $\equiv \dots$

## Specifying *Image* operations

$\mu$  (*over top bot*)  $\equiv \lambda p \rightarrow \text{overC } (\mu \text{ top } p) (\mu \text{ bot } p)$   
 $\mu$  (*crop reg im*)  $\equiv \lambda p \rightarrow \mathbf{if} \ \mu \text{ reg } p \ \mathbf{then} \ \mu \text{ im } p \ \mathbf{else} \ \text{clear}$   
 $\mu$  (*monochrome c*)  $\equiv \lambda p \rightarrow c$   
 $\mu$  (*transform tr im*)  $\equiv \quad \text{-- coming up}$

$\text{overC} :: \text{Color} \rightarrow \text{Color} \rightarrow \text{Color}$

Note compositionality of  $\mu$ .



# Compositional semantics

Make more explicit:

$$\mu (\textit{over top bot}) \equiv \textit{overS} (\mu \textit{top}) (\mu \textit{bot})$$
$$\mu (\textit{crop reg im}) \equiv \textit{cropS} (\mu \textit{reg}) (\mu \textit{im})$$
$$\textit{overS} :: (\textit{Loc} \rightarrow \textit{Color}) \rightarrow (\textit{Loc} \rightarrow \textit{Color}) \rightarrow (\textit{Loc} \rightarrow \textit{Color})$$
$$\textit{overS} f g = \lambda p \rightarrow \textit{overC} (f p) (g p)$$
$$\textit{cropS} :: (\textit{Loc} \rightarrow \textit{Bool}) \rightarrow (\textit{Loc} \rightarrow \textit{Color}) \rightarrow (\textit{Loc} \rightarrow \textit{Color})$$
$$\textit{cropS} f g = \lambda p \rightarrow \mathbf{if} f p \mathbf{then} g p \mathbf{else} \textit{clear}$$

## Generalize and simplify

---

- What about transforming *regions*?
- Other pointwise combinations (lerp, threshold)?

Generalize:

**type** *Image* *a*

**type** *ImageC* = *Image Color*

**type** *Region* = *Image Bool*

Now some operations become more general.

## Generalize and simplify

$transform :: Transform \rightarrow Image\ a \rightarrow Image\ a$

$cond \quad \quad :: Image\ Bool \rightarrow Image\ a \rightarrow Image\ a \rightarrow Image\ a$

$lift_0 :: a \rightarrow Image\ a$

$lift_1 :: (a \rightarrow b) \rightarrow (Image\ a \rightarrow Image\ b)$

$lift_2 :: (a \rightarrow b \rightarrow c) \rightarrow (Image\ a \rightarrow Image\ b \rightarrow Image\ c)$

...

Specializing,

$monochrome = lift_0$

$over \quad \quad = lift_2\ overC$

$crop\ r\ im \quad = cond\ r\ im\ emptyIm$

$cond \quad \quad = lift_3\ ifThenElse$

# Spatial transformation

---

$\mu :: \textit{Transform} \rightarrow ??$

$\mu (\textit{transform tr im}) \equiv ??$

# Spatial transformation

$\mu :: \text{Transform} \rightarrow ??$

$\mu (\text{transform } tr \text{ im}) \equiv \text{transformS } (\mu \text{ tr}) (\mu \text{ im})$

where

$\text{transformS} :: ?? \rightarrow (\text{Loc} \rightarrow \text{Color}) \rightarrow (\text{Loc} \rightarrow \text{Color})$

# Spatial transformation

$$\mu :: \text{Transform} \rightarrow (\text{Loc} \rightarrow \text{Loc})$$
$$\mu (\text{transform } tr \text{ im}) \equiv \text{transformS } (\mu \text{ tr}) (\mu \text{ im})$$

where

$$\text{transformS} :: (\text{Loc} \rightarrow \text{Loc}) \rightarrow (\text{Loc} \rightarrow \text{Color}) \rightarrow (\text{Loc} \rightarrow \text{Color})$$
$$\text{transformS } h \text{ f} = \lambda p \rightarrow \text{f } (h \text{ p})$$

Subtle implications.

What is *Loc*? My answer: continuous, infinite 2D space.

**type** *Loc* =  $\mathbb{R}^2$

# Why continuous & infinite (vs discrete/finite) space?

Same benefits as for time (FRP):

- Transformation flexibility with simple & precise semantics.
- Modularity/reusability/composability:
  - Fewer assumptions, more uses (resolution-independence).
  - More info available for extraction.
- Integration and differentiation: natural, accurate, efficient.
- Quality/accuracy.
- Efficiency (adapative).
- Reconcile differing input sampling rates.

*Principle: Approximations/prunings compose badly, so postpone.*

See *Why Functional Programming Matters*.

Pan gallery



## Using standard vocabulary

---

- We've created a domain-specific vocabulary.
- Can we reuse standard vocabularies instead?
- Why would we want to?
  - User knowledge.
  - Ecosystem support (multiplicative power).
  - Laws as sanity check.
  - Tao check.
  - Specification and laws for free, as we'll see.
- In Haskell, standard type classes.

# Monoid

Interface:

**class** *Monoid* *m* **where**

$\varepsilon$  :: *m* -- “mempty”

$(\oplus)$  :: *m* → *m* → *m* -- “mappend”

Laws:

$$a \oplus \varepsilon \equiv a$$

$$\varepsilon \oplus b \equiv b$$

$$a \oplus (b \oplus c) \equiv (a \oplus b) \oplus c$$

Why do laws *matter*? Compositional (modular) reasoning.

What monoids have we seen today?

# Image monoid

**instance** *Monoid ImageC* **where**

$\varepsilon = \text{lift}_0 \text{ clear}$

$(\oplus) = \text{over}$

Is there a more general form on *Image a*?

**instance** *Monoid a*  $\Rightarrow$  *Monoid (Image a)* **where**

$\varepsilon = \text{lift}_0 \varepsilon$

$(\oplus) = \text{lift}_2 (\oplus)$

Do these instances satisfy the *Monoid* laws?

# Functor

---

```
class Functor f where  
  fmap :: (a → b) → (f a → f b)
```

For images?

```
instance Functor Image where  
  fmap = lift1
```

Laws?

## Applicative

**class** *Functor*  $f \Rightarrow$  *Applicative*  $f$  **where**

*pure*  $:: a \rightarrow f\ a$

$(\langle * \rangle) :: f\ (a \rightarrow b) \rightarrow f\ a \rightarrow f\ b$

For images?

**instance** *Applicative* *Image* **where**

*pure* = *lift*<sub>0</sub>

$(\langle * \rangle) = \text{lift}_2\ (\$)$

From *Applicative*, where  $(\langle \$ \rangle) = \text{fmap}$ :

$\text{lift}A_2\ f\ p\ q = f\ \langle \$ \rangle\ p\ \langle * \rangle\ q$

$\text{lift}A_3\ f\ p\ q\ r = f\ \langle \$ \rangle\ p\ \langle * \rangle\ q\ \langle * \rangle\ r$

-- etc

Laws?

# Instance semantics

*Monoid:*

$$\mu \varepsilon \quad \equiv \lambda p \rightarrow \varepsilon$$

$$\mu (top \oplus bot) \equiv \lambda p \rightarrow \mu top p \oplus \mu bot p$$

*Functor:*

$$\mu (fmap f im) \equiv \lambda p \rightarrow f (\mu im p)$$

$$\equiv f \circ \mu im$$

*Applicative:*

$$\mu (pure a) \quad \equiv \lambda p \rightarrow a$$

$$\mu (imf \langle * \rangle imx) \equiv \lambda p \rightarrow (\mu imf p) (\mu imx p)$$

## Monad and Comonad

**class** *Monad* *f* **where**

*return* ::  $a \rightarrow f\ a$

*join* ::  $f\ (f\ a) \rightarrow f\ a$

**class** *Functor* *f*  $\Rightarrow$  *Comonad* *f* **where**

*coreturn* ::  $f\ a \rightarrow a$

*cojoin* ::  $f\ a \rightarrow f\ (f\ a)$

*Comonad* gives us neighborhood operations.

# Monoid specification, revisited

Image monoid specification:

$$\mu \varepsilon \quad \equiv \lambda p \rightarrow \varepsilon$$

$$\mu (top \oplus bot) \equiv \lambda p \rightarrow \mu top p \oplus \mu bot p$$

Instance for the semantic model:

**instance** *Monoid*  $m \Rightarrow$  *Monoid* ( $z \rightarrow m$ ) **where**

$$\varepsilon \quad = \lambda z \rightarrow \varepsilon$$

$$f \oplus g = \lambda z \rightarrow f z \oplus g z$$

Refactoring,

$$\mu \varepsilon \quad \equiv \varepsilon$$

$$\mu (top \oplus bot) \equiv \mu top \oplus \mu bot$$

So  $\mu$  *distributes* over monoid operations, i.e., a monoid homomorphism.



# Functor specification, revisited

Functor specification:

$$\mu (fmap f im) \equiv f \circ \mu im$$

Instance for the semantic model:

**instance** *Functor* (( $\rightarrow$ ) *u*) **where**

$$fmap f h = f \circ h$$

Refactoring,

$$\mu (fmap f im) \equiv fmap f (\mu im)$$

So  $\mu$  is a *functor* homomorphism.

# Applicative specification, revisited

Applicative specification:

$$\mu (\text{pure } a) \quad \equiv \lambda p \rightarrow a$$

$$\mu (\text{imf } \langle * \rangle \text{ imx}) \equiv \lambda p \rightarrow (\mu \text{ imf } p) (\mu \text{ imx } p)$$

Instance for the semantic model:

**instance** *Applicative* (( $\rightarrow$ ) *u*) **where**

$$\text{pure } a \quad = \lambda u \rightarrow a$$

$$\text{fs } \langle * \rangle \text{ xs} = \lambda u \rightarrow (\text{fs } u) (\text{xs } u)$$

Refactoring,

$$\mu (\text{pure } a) \quad \equiv \text{pure } a$$

$$\mu (\text{imf } \langle * \rangle \text{ imx}) \equiv \mu \text{ imf } \langle * \rangle \mu \text{ imx}$$

So  $\mu$  is an *applicative* homomorphism.

# Specifications for free

---

Semantic type class morphism (TCM) principle:

*The instance's meaning follows the meaning's instance.*

That is, the type acts like its meaning.

Every TCM failure is an abstraction leak.

Strong design principle.

Class laws necessarily hold, as we'll see.

## Laws for free

$$\begin{array}{l} \mu \varepsilon \quad \equiv \varepsilon \\ \mu (a \oplus b) \equiv \mu a \oplus \mu b \end{array} \Rightarrow \begin{array}{l} a \oplus \varepsilon \quad \equiv a \\ \varepsilon \oplus b \quad \equiv b \\ a \oplus (b \oplus c) \equiv (a \oplus b) \oplus c \end{array}$$

where equality is *semantic*. Proofs:

$\begin{array}{l} \mu (a \oplus \varepsilon) \\ \equiv \mu a \oplus \mu \varepsilon \\ \equiv \mu a \oplus \varepsilon \\ \equiv \mu a \end{array}$	$\begin{array}{l} \mu (\varepsilon \oplus b) \\ \equiv \mu \varepsilon \oplus \mu b \\ \equiv \varepsilon \oplus \mu b \\ \equiv \mu b \end{array}$	$\begin{array}{l} \mu (a \oplus (b \oplus c)) \\ \equiv \mu a \oplus (\mu b \oplus \mu c) \\ \equiv (\mu a \oplus \mu b) \oplus \mu c \\ \equiv \mu ((a \oplus b) \oplus c) \end{array}$
---	---	--

Works for other classes as well.

# Example: functional reactive programming

---

See previous talks:

- *The essence and origins of FRP*
- *A more elegant specification for FRP*

## Example: uniform pairs

---

Type:

```
data Pair a = a :# a
```

API: *Monoid*, *Functor*, *Applicative*, *Monad*, *Foldable*, *Traversable*.

Specification follows from simple & precise denotation.

## Uniform pairs — denotation

*Pair* is an *indexable* container. What's the index type?

**type**  $P\ a = \text{Bool} \rightarrow a$

$\mu :: \text{Pair}\ a \rightarrow P\ a$

$\mu\ (u \text{ :\# } v)\ \text{False} = u$

$\mu\ (u \text{ :\# } v)\ \text{True} = v$

API specification? Homomorphisms, as usual!

# Uniform pairs — monoid

Monoid homomorphism:

$$\mu \varepsilon \quad \equiv \quad \varepsilon$$

$$\mu (u \oplus v) \equiv \mu u \oplus \mu v$$

In this case,

**instance** *Monoid*  $m \Rightarrow$  *Monoid*  $(z \rightarrow m)$  **where**

$$\varepsilon \quad = \quad \lambda z \rightarrow \varepsilon$$

$$f \oplus g = \lambda z \rightarrow f z \oplus g z$$

so

$$\mu \varepsilon \quad \equiv \quad \lambda z \rightarrow \varepsilon$$

$$\mu (u \oplus v) \equiv \lambda z \rightarrow \mu u z \oplus \mu v z$$

Implementation: solve for  $\varepsilon$  and  $(\oplus)$  on the left. Hint: find  $\mu^{-1}$ .



# Uniform pairs — other classes

---

Exercise: apply the same principle for

- *Functor*
- *Applicative*
- *Monad*
- *Foldable*
- *Traversable*

## Example: streams

**data** *Stream* *a* = *Cons* *a* (*Stream* *a*)

API: same classes as with *Pair*.

Denotation? Hint: *Stream* is also an indexable type.

**data** *S* *a* = *Nat*  $\rightarrow$  *a*

**data** *Nat* = *Zero* | *Succ* *Nat*

Interpret *Stream* as *S*:

$\mu :: \textit{Stream } a \rightarrow \textit{S } a$

$\mu (\textit{Cons } a \textit{ } \_) \textit{ Zero} = a$

$\mu (\textit{Cons } \_ \textit{ as}) (\textit{Succ } n) = \mu \textit{ as } n$

# Memo tries

Generalizes *Pair* and *Stream*:

**type**  $a \rightarrow b$

$\mu :: (a \rightarrow b) \rightarrow (a \rightarrow b)$

API: classes as above, plus *Category*.

Exploit inverses to calculate instances, e.g.,

$$\begin{aligned} \mu \text{ id} &\equiv \text{id} \\ \Leftarrow \text{id} &\equiv \mu^{-1} \text{ id} \end{aligned}$$

$$\begin{aligned} \mu (g \circ f) &\equiv \mu g \circ \mu f \\ \Leftarrow g \circ f &\equiv \mu^{-1} (\mu g \circ \mu f) \end{aligned}$$

Then simplify/optimize.

## Example: lists with a bonus

---

**data** *ListX a b = Done b | Cons a (ListX a b)*

Denotation:

$\mu :: \text{ListX } a \ b \rightarrow ([a], b)$

$\mu (\text{Done } b) = ([], b)$

$\mu (\text{Cons } a \ asb) = (a : as, b)$  **where**  $(as, b) = \mu \ asb$

Exercise: instances, including

**instance** *Monad (ListX a) where ...*

Then generalize from lists to arbitrary monoid.

# Example: linear transformations

---

*Assignment:*

- Represent linear transformations
- Scalar, non-scalar domain & range, identity and composition

*Plan:*

- Interface
- Denotation
- Representation
- Calculation (implementation)

## Interface and denotation

Interface:

$$\begin{aligned} \mathbf{type} \quad (& \multimap) :: * \rightarrow * \rightarrow * \\ \mathit{scale} &:: \mathit{Num} \, s \Rightarrow (s \multimap s) \\ \widehat{\mathit{id}} &:: a \multimap a \\ (\widehat{\circ}) &:: (b \multimap c) \rightarrow (a \multimap b) \rightarrow (a \multimap c) \\ &\dots \end{aligned}$$

Model:

$$\begin{aligned} \mathbf{type} \quad a \multimap b & \text{ -- Linear subset of } a \rightarrow b \\ \mu &:: (a \multimap b) \rightarrow (a \rightarrow b) \end{aligned}$$

Specification:

$$\begin{aligned} \mu (\mathit{scale} \, s) &\equiv \lambda x \rightarrow s \times x \\ \mu \widehat{\mathit{id}} &\equiv \mathit{id} \\ \mu (g \widehat{\circ} f) &\equiv \mu g \circ \mu f \\ &\dots \end{aligned}$$

# Representation

Start with 1D. Recall partial specification:

$$\mu (\text{scale } s) \equiv \lambda x \rightarrow s \times x$$

Try a direct data type representation:

```
data ( $\text{:-}\circ$ ) :: *  $\rightarrow$  *  $\rightarrow$  * where  
  Scale :: Num s  $\Rightarrow$  s  $\rightarrow$  (s  $\text{:-}\circ$  s)  -- ...  
 $\mu$  :: (a  $\text{:-}\circ$  b)  $\rightarrow$  (a  $\text{-}\circ$  b)  
 $\mu$  (Scale s) =  $\lambda x \rightarrow s \times x$ 
```

Spec trivially satisfied by  $\text{scale} = \text{Scale}$ .

Others are more interesting.

# Calculate an implementation

Specification:

$$\mu \hat{id} \equiv id$$

$$\mu (g \hat{\circ} f) \equiv \mu g \circ \mu f$$

Calculation:

$$\begin{aligned} id \\ \equiv \lambda x \rightarrow x \\ \equiv \lambda x \rightarrow 1 \times x \\ \equiv \mu (Scale\ 1) \end{aligned}$$

$$\begin{aligned} \mu (Scale\ s) \circ \mu (Scale\ s') \\ \equiv (\lambda x \rightarrow s \times x) \circ (\lambda x' \rightarrow s' \times x') \\ \equiv \lambda x' \rightarrow s \times (s' \times x') \\ \equiv \lambda x' \rightarrow ((s \times s') \times x') \\ \equiv \mu (Scale\ (s \times s')) \end{aligned}$$

Sufficient definitions:

$$\hat{id} = Scale\ 1$$

$$Scale\ s \hat{\circ} Scale\ s' = Scale\ (s \times s')$$



# Algebraic abstraction

---

In general,

- Replace ad hoc vocabulary with a standard abstraction.
- Recast semantics as homomorphism.
- Note that laws hold.

What standard abstraction to use for  $(:-\circ)$ ?

# Category

Interface:

```
class Category k where  
  id :: k a a  
  (o) :: k b c → k a b → k a c
```

Laws:

$$\begin{aligned} id \circ f &\equiv f \\ g \circ id &\equiv g \\ (h \circ g) \circ f &\equiv h \circ (g \circ f) \end{aligned}$$

# Linear transformation category

Linear map semantics:

$$\mu :: (a \multimap b) \rightarrow (a \multimap b)$$

$$\mu (\text{Scale } s) = \lambda x \rightarrow s \times x$$

Specification as homomorphism (no abstraction leak):

$$\mu \text{ id} \quad \equiv \text{ id}$$

$$\mu (g \circ f) \equiv \mu g \circ \mu f$$

Correct-by-construction implementation:

**instance** *Category* ( $\multimap$ ) **where**

$$\text{id} = \text{Scale } 1$$

$$\text{Scale } s \circ \text{Scale } s' = \text{Scale } (s \times s')$$

# Laws for free

$\begin{aligned}\mu id &\equiv id \\ \mu (g \circ f) &\equiv \mu g \circ \mu f\end{aligned}$	$\Rightarrow$	$\begin{aligned}id \circ f &\equiv f \\ g \circ id &\equiv g \\ (h \circ g) \circ f &\equiv h \circ (g \circ f)\end{aligned}$
--	---------------	---

where equality is *semantic*. Proofs:

$\begin{aligned}\mu (id \circ f) \\ &\equiv \mu id \circ \mu f \\ &\equiv id \circ \mu f \\ &\equiv \mu f\end{aligned}$	$\begin{aligned}\mu (g \circ id) \\ &\equiv \mu g \circ \mu id \\ &\equiv \mu g \circ id \\ &\equiv \mu g\end{aligned}$	$\begin{aligned}\mu ((h \circ g) \circ f) \\ &\equiv (\mu h \circ \mu g) \circ \mu f \\ &\equiv \mu h \circ (\mu g \circ \mu f) \\ &\equiv \mu (h \circ (g \circ f))\end{aligned}$
---	---	--

Works for other classes as well.

# Higher dimensions

Interface:

$$(\triangle) :: (a : \multimap c) \rightarrow (a : \multimap d) \rightarrow (a : \multimap c \times d)$$

$$(\nabla) :: (a : \multimap c) \rightarrow (b : \multimap c) \rightarrow (a \times b : \multimap c)$$

Semantics:

$$\mu (f \triangle g) \equiv \lambda a \rightarrow (f a, g a)$$

$$\mu (f \nabla g) \equiv \lambda(a, b) \rightarrow f a + g b$$

# Products and coproducts

**class** *Category*  $k \Rightarrow$  *ProductCat*  $k$  **where**

**type**  $a \times_k b$

$exl :: k (a \times_k b) a$

$exr :: k (a \times_k b) b$

$(\Delta) :: k a c \rightarrow k a d \rightarrow k a (c \times_k d)$

**class** *Category*  $k \Rightarrow$  *CoproductCat*  $k$  **where**

**type**  $a +_k b$

$inl :: k a (a +_k b)$

$inr :: k b (a +_k b)$

$(\nabla) :: k a c \rightarrow k b c \rightarrow k (a +_k b) c$

Similar to *Arrow* and *ArrowChoice* classes.

## Semantic morphisms

$$\mu \text{ exl} \quad \equiv \text{exl}$$

$$\mu \text{ exr} \quad \equiv \text{exr}$$

$$\mu (f \triangle g) \equiv \mu f \triangle \mu g$$

$$\mu \text{ inl} \quad \equiv \text{inl}$$

$$\mu \text{ inr} \quad \equiv \text{inr}$$

$$\mu (f \nabla g) \equiv \mu f \nabla \mu g$$

For  $a \multimap b$ ,

$$\mathbf{type} \ a \times_{(\multimap)} b = a \times b$$

$$\text{exl} \ (a, b) = a$$

$$\text{exr} \ (a, b) = b$$

$$f \triangle g = \lambda a \rightarrow (f \ a, g \ a)$$

$$\mathbf{type} \ a +_{(\multimap)} b = a \times b$$

$$\text{inl} \ a = (a, 0)$$

$$\text{inr} \ b = (0, b)$$

$$f \nabla g = \lambda(a, b) \rightarrow f \ a + g \ b$$

For calculation, see blog post *Reimagining matrices*.

# Full representation and denotation

**data**  $(:-\circ) :: * \rightarrow * \rightarrow *$  **where**

*Scale*  $:: \text{Num } s \Rightarrow s \rightarrow (s :-\circ s)$

$(:\triangle) :: (a :-\circ c) \rightarrow (a :-\circ d) \rightarrow (a :-\circ c \times d)$

$(:\nabla) :: (a :-\circ c) \rightarrow (b :-\circ c) \rightarrow (a \times b :-\circ c)$

$\mu :: (a :-\circ b) \rightarrow (a \text{ } \text{---} \text{ } b)$

$\mu (\text{Scale } s) = \lambda x \rightarrow s \times x$

$\mu (f :\triangle g) = \lambda a \rightarrow (f a, g a)$

$\mu (f :\nabla g) = \lambda(a, b) \rightarrow f a + g b$



# Denotational design

---

Design methodology for typed, purely functional programming:

- Precise, simple, and compelling specification.
- Informs *use* and *implementation* without entangling.
- Standard algebraic abstractions.
- Free of abstraction leaks.
- Laws for free.
- Principled construction of correct implementation.

# References

---

- *Denotational design with type class morphisms*
- *Push-pull functional reactive programming*
- Functional images (Pan) page with pictures & papers.
- Posts on type class morphisms
- *Reimagining matrices*
- This workshop